Development and application of a multi-route physiologically based pharmacokinetic model for oxytetracycline in dogs and humans

Zhoumeng Lin, Mengjie Li, Ronette Gehring, Jim E. Riviere*

Institute of Computational Comparative Medicine (ICCM) and The Department of Anatomy and Physiology, College of Veterinary Medicine, Kansas State University, Manhattan, KS 66506, USA

J Pharm Sci. 2014 Nov 18. doi: 10.1002/jps.24244. [Epub ahead of print]

* Corresponding author: ICCM, Department of Anatomy and Physiology, College of Veterinary Medicine, Kansas State University, 1800 Denison Avenue, P200 Mosier Hall, Manhattan, KS 66506, USA. E-mail: jriviere@ksu.edu; phone: +1-785-532-3683; fax: +1-785-532-4557.

Model code:

```
PROGRAM
! Initiated by Zhoumeng Lin on 02/11/2014;
! Last checked by Zhoumeng Lin on 07/28/2014;

INITIAL
! code that is executed once at the beginning of a simulation run goes here
!! Physiological parameters
! Blood flow rates
 CONSTANT QCC = 12.9 ! Cardiac output (L/h/kg) (Brown et al., 1997, Table 22 for mixed-breed (Mongrel) dogs (8.39 L/h/kg) and beagles (12.9 L/h/kg))
 CONSTANT QLC = 0.297 ! Fraction of blood flow to the liver (Brown et al., 1997, Table 26)
 CONSTANT QKC = 0.173 ! Fraction of blood flow to the kidneys (Brown et al., 1997, Table 26)
 CONSTANT QFC = 0.097 ! Fraction of blood flow to the fat (Vinegar, 2001, Table 2)
 CONSTANT QMC = 0.217 ! Fraction of blood flow to the muscle (Brown et al., 1997, Table 26)

! Tissue volumes
 CONSTANT BW = 11.3 ! Body weight (kg) (Baggot et al., 1977)
 CONSTANT VLC = 0.0329 ! Fractional liver tissue (Brown et al., 1997, Table 6)
 CONSTANT VKC = 0.0055 ! Fractional kidney tissue (Brown et al., 1997, Table 6)
 CONSTANT VFC = 0.15 ! Fractional fat tissue (Vinegar, 2001, Table 2)
 CONSTANT VMC = 0.4565 ! Fractional muscle tissue (Brown et al., 1997, Table 6)
 CONSTANT VbloodC = 0.082 ! Blood volume, fraction of BW (Brown et al., 1997, Table 21)

!Fraction of tissue volumes that is blood ! (Brown et al., 1997; Table 30)
 CONSTANT FVBF = 0.02 ! Blood volume fraction of fat (%)
 CONSTANT FVBM = 0.01 ! Blood volume fraction of muscle (%)
 CONSTANT FVBS = 0.01 ! Blood volume fraction of slowly perfused tissues (%)

! Mass Transfer Parameters (Chemical-specific parameters)
! Partition coefficients (PC, tissue:plasma)
 CONSTANT PL = 1.89 ! Liver:plasma PC (Craigmill et al., 2000, Table 3; Craigmill, 2003, Table 4, in sheep)
 CONSTANT PK = 4.75 ! Kidney:plasma PC (Craigmill et al., 2000, Table 3; Craigmill, 2003, Table 4, in sheep)
 CONSTANT PM = 0.85 ! Muscle:plasma PC (Craigmill et al., 2000, Table 3; Craigmill, 2003, Table 4, in sheep)
 CONSTANT PF = 0.086 ! Fat:plasma PC (Craigmill et al., 2000, Table 3; Craigmill, 2003, Table 4, in sheep)
 CONSTANT PR = 4.75 ! Richly perfused tissues:plasma PC (Assumed the same as kidney:plasma PC)
 CONSTANT PS = 0.85 ! Slowly perfused tissues:plasma PC (Assumed the same as muscle:plasma PC)

!Permeability constants (L/h/kg tissue) (Permeation area cross products)
 CONSTANT PAFC = 0.012 ! Fat tissue permeability (0.12 in Leavens et al., 2012)
 CONSTANT PAMC = 0.225 ! Muscle tissue permeability (0.45 in Leavens et al., 2012)
 CONSTANT PASC = 0.049 ! Slowly perfused tissue permeability (0.49 in Leavens et al., 2012)

! Kinetic constants
```

```
! Oral absorption rate constants
 CONSTANT Kst = 2 ! /h, gastric emptying rate constant
 CONSTANT Ka = 0.012 ! /h, intestinal absorption rate constant, ka=0.05 for experimental solution; ka=0.012 for tablets or capsules
 CONSTANT Kint = 0.2 ! /h, intestinal transit rate constant

! IM absorption rate constants
 CONSTANT Kim = 0.3 ! 0.15 for conventional formulation; 0.3 for long-acting formulation; IM absorption rate constant (/h)
 CONSTANT Frac = 0.5 ! 0.95 for conventional formulation; 0.5 for long-acting formulation
 CONSTANT Kdiss = 0.02 ! /h

! IV infusion rate constants
 CONSTANT Timeiv = 0.01 ! IV infusion time (h), based on Leavens et al., 2012

 ! Urinary elimination rate constant
  CONSTANT KurineC = 0.2 ! L/h/kg

!Parameters for various exposure scenarios
 Constant PDOSEoral = 0 ! (mg/kg)
 Constant PDOSEiv = 10 ! (mg/kg)
 Constant PDOSEim = 0 ! (mg/kg)

END ! INITIAL

DYNAMIC

ALGORITHM IALG = 2
NSTEPS    NSTP = 10
MAXTERVAL MAXT = 1.0e9
MINTERVAL MINT = 1.0e-9
CINTERVAL CINT = 0.1

DERIVATIVE
! code for calculating the derivative goes here
! Cardiac output and blood flows to tissues (L/h)
 QC = QCC*BW ! Cardiac output
 QL = QLC*QC ! Liver
 QK = QKC*QC ! Kidney
 QF = QFC*QC ! Fat
 QM = QMC*QC ! Muscle
 QR = 0.626*QC-QK-QL ! Richly perfused tissues
 QS = 0.374*QC-QF-QM ! Slowly perfused tissues

! Tissue volumes (L)
 VL = VLC*BW ! Liver
 VK = VKC*BW ! Kidney
 VF = VFC*BW ! Fat
 VM = VMC*BW ! Muscle
 Vblood = VbloodC*BW ! Blood
 VR = 0.142*BW-VL-VK ! Richly perfused tissues
```

```
 VS = 0.776*BW-VF-VM ! Slowly perfused tissues

!Permeability surface area coefficients
 PAF = PAFC*VF ! Fat:blood permeability (L/h)
 PAM = PAMC*VM ! Muscle:blood permeability (L/h)
 PAS = PASC*VS ! Slowly perfused tissue:blood permeability (L/h)

! volume of tissue vs blood
! Fat
 VFb = FVBF*VF ! Fat compartment blood volume
 VFt = VF-VFb ! Fat compartment tissue volume

! Muscle
 VMb = FVBM*VM ! Muscle compartment blood volume
 VMt = VM-VMb ! Muscle compartment tissue volume

! Slowly perfused tissue
 VSb = FVBS*VS ! Slowly perfused compartment blood volume
 VSt = VS-VSb ! Slowly perfused compartment tissue volume

! Dosing
 DOSEoral = PDOSEoral*BW ! (mg)
 DOSEiv = PDOSEiv*BW ! (mg)
 DOSEim = PDOSEim*BW ! (mg)

!...Dosing, multiple oral gavage
 CONSTANT tlen = 0.001 ! Length of oral gavage exposure (h/day)
 CONSTANT tinterval = 6 ! Varied dependent on the exposure paradigm
 CONSTANT Dstart = 0.0 ! Initiation day of oral gavage (day)
 CONSTANT Dstop = 0.2 ! Termination day of oral gavage (day)
 CONSTANT MAXT = 1.0  ! maximum comm. interval
 CONSTANT CINTC = 0.1 ! Communication interval
 CINT = CINTC  ! Communication interval

 Tsim = TSTOP ! Tstop in hours
 DS = Dstart*24  ! Initiation time point of oral gavage (h)
 Doff = (Dstop - Dstart)*24 ! Oral gavage duration (h)

 TimeOn = Dstart*24! Initiation time point of oral gavage (h)
 TimeOff = Dstop*24+tlen ! Termination time point of oral gavage (h)

 Exposure = PULSE(0,tinterval,tlen)*PULSE(DS, Tsim, Doff)
 Rdoseoral = (Doseoral/tlen)*Exposure
 RAST = RDoseoral-Kst*AST
 AST = Integ(RAST,0)
 RAI = Kst*AST-Ka*AI-Kint*AI
 Rcolon=Kint*AI
 Acolon=Integ(Rcolon,0)
 AI = Integ(RAI,0)
 RAO=Ka*AI
 AAO = Integ(RAO,0)
```

```
!...Dosing, intramuscular, dissolution model
 Doseimfast = Doseim*Frac
 Doseimslow = Doseim*(1-Frac)

 Rim=Kim*Amtsite
 Absorb=Integ(Rim,0)
 Rsite=-Rim+Kdiss*Doseimremain
 Amtsite=Integ(Rsite,Doseimfast)
 Rdoseimremain=-Kdiss*Doseimremain
 Doseimremain=Integ(Rdoseimremain,Doseimslow)

! OTC iv injection to the venous
 IVR = DOSEiv/timeiv
 RIV = IVR*(1.-step(timeiv))
 Aiv = Integ(Riv,0)

! Urinary elimination rate constant
 Kurine = KurineC*BW

! OTC in blood compartment
 CV = ((QL*CVL+QK*CVK+QF*CVF+QM*CVM+QR*CVR+QS*CVS+Riv+Rim)/QC)
 RA = QC*(CV-CA)
 AA = Integ(RA,0)
 CA = AA/Vblood
 AUCCV = Integ(CV,0.0)

! OTC in liver compartment
 RL = QL*(CA-CVL)+RAO
 AL = Integ(RL,0)
 CL = AL/VL
 CVL = AL/(VL*PL)
 AUCCL = Integ(CL,0.0)

! OTC in kidney compartment
 RK = QK*(CA-CVK)-Rurine
 AK = Integ(RK,0)
 CK = AK/VK
 CVK = AK/(VK*PK)
 AUCCK = Integ(CK,0.0)

! Urinary excretion of OTC
 Rurine = Kurine*CVK
 Aurine = Integ(Rurine,0)

! OTC in fat compartment, permeability-limited model
 RFB = QF*(CA-CVF)-PAF*CVF+(PAF*CFt)/PF
 AFB = Integ(RFB,0)
 CVF = AFB/VFB

 RFt = PAF*CVF-(PAF*CFt)/PF
```

```
 AFt = Integ(RFt,0)
 CFt = AFt/VFT
 Aftotal = Aft+Afb
 CF = Aftotal/VF

!! OTC in muscle compartment, permeability-limited model
 RMB = QM*(CA-CVM)-PAM*CVM+(PAM*CMt)/PM
 AMB = Integ(RMB,0)
 CVM = AMB/VMB

 RMt = PAM*CVM-(PAM*CMt)/PM
 AMt = Integ(RMt,0)
 CMt = AMt/VMT
 AMtotal = AMt+AMb
 CM = AMtotal/VM
 AUCCM = Integ(CM,0.0)

! OTC in richly perfused tissue compartment
 RR = QR*(CA-CVR)
 AR = Integ(RR,0)
 CR = AR/VR
 CVR = AR/(VR*PR)

! OTC in slowly perfused tissue compartment, permeability-limited model
 RSB = QS*(CA-CVS)-PAS*CVS+(PAS*CSt)/PS
 ASB = Integ(RSB,0)
 CVS = ASB/VSB

 RSlt = PAS*CVS-(PAS*CSt)/PS
 ASlt = Integ(RSlt,0)
 CSt = ASlt/VST
 AStotal = ASlt+ASb
 CS = AStotal/VS

! Mass balance
Qbal = QC-QL-QK-QF-QM-QR-QS
Tmass = AA+AL+AK+AFtotal+AMtotal+AR+AStotal+Aurine
Bal = AAO+Aiv+Absorb-Tmass ! Permeability-limited model mass balance

END ! DERIVATIVE

! Add discrete events here as needed
!DISCRETE
!END

! code that is executed once at each communication interval goes here

CONSTANT TSTOP = 24.0
TERMT (T .GE. TSTOP, 'checked on communication interval: REACHED TSTOP')

END ! DYNAMIC
```

```
TERMINAL
! code that is executed once at the end of a simulation run goes here
END ! TERMINAL

END ! PROGRAM
```